Quasi-Homogenes Aufblasen einer Spitze

Polares Aufblasen

```
f := matrix(2,1,[ y, x^2]);
```

$$\begin{pmatrix} y \\ x^2 \end{pmatrix}$$

```
a := 2: b := 3:
Psi := [ r^a*cos(phi), r^b*sin(phi) ];
SPsi := [ x=r^a*cos(phi), y=r^b*sin(phi) ]:
J := jacobian(Psi, [r,phi]);
JI := J^(-1)
```

$$[r^2 \cos(\mathrm{phi}), r^3 \sin(\mathrm{phi})]$$

$$\begin{pmatrix} 2\,r \cos(\mathrm{phi}) & -r^2 \sin(\mathrm{phi}) \\ 3\,r^2 \sin(\mathrm{phi}) & r^3 \cos(\mathrm{phi}) \end{pmatrix}$$

$$\begin{pmatrix} \frac{\cos(\mathrm{phi})}{2\,r \cos(\mathrm{phi})^2 + 3\,r \sin(\mathrm{phi})^2} & \frac{\sin(\mathrm{phi})}{\sigma_1} \\ -\frac{3 \sin(\mathrm{phi})}{\sigma_1} & \frac{2 \cos(\mathrm{phi})}{2\,r^3 \cos(\mathrm{phi})^2 + 3\,r^3 \sin(\mathrm{phi})^2} \end{pmatrix}$$

where

$$\sigma_1 = 2\,r^2 \cos(\mathrm{phi})^2 + 3\,r^2 \sin(\mathrm{phi})^2$$

```
g := map(map(JI*subs(f,SPsi),expand),collect,r); g := g/r; simplify(g)
```

$$\begin{pmatrix} \frac{\left(\sin(\mathrm{phi}) \cos(\mathrm{phi})^2 + \sin(\mathrm{phi}) \cos(\mathrm{phi})\right) r^2}{2 \cos(\mathrm{phi})^2 + 3 \sin(\mathrm{phi})^2} \\ \frac{\left(2 \cos(\mathrm{phi})^3 - 3 \sin(\mathrm{phi})^2\right) r}{2 \cos(\mathrm{phi})^2 + 3 \sin(\mathrm{phi})^2} \end{pmatrix}$$

$$\begin{pmatrix} \frac{r \left(\sin(\mathrm{phi}) \cos(\mathrm{phi})^2 + \sin(\mathrm{phi}) \cos(\mathrm{phi})\right)}{2 \cos(\mathrm{phi})^2 + 3 \sin(\mathrm{phi})^2} \\ \frac{2 \cos(\mathrm{phi})^3 - 3 \sin(\mathrm{phi})^2}{2 \cos(\mathrm{phi})^2 + 3 \sin(\mathrm{phi})^2} \end{pmatrix}$$

$$\begin{pmatrix} -\frac{r \sin(2\,\mathrm{phi}) (\cos(\mathrm{phi}) + 1)}{2 \left(\cos(\mathrm{phi})^2 - 3\right)} \\ -\frac{2 \cos(\mathrm{phi})^3 + 3 \cos(\mathrm{phi})^2 - 3}{\cos(\mathrm{phi})^2 - 3} \end{pmatrix}$$

```
nsol := numeric::solve(numer(g[2]), phi=-PI..PI, AllRealRoots)
```

$$\{-0.6326829902, 0.6326829902\}$$

```
JJ := simplify(jacobian(g, [r,phi])):
nsol[1]; JJ1 := simplify(subs(JJ, r=0, phi=nsol[1]));
nsol[2]; JJ2 := simplify(subs(JJ, r=0, phi=nsol[2]));
```

$$-0.6326829902$$

$$\begin{pmatrix} -0.3666160417 & 0 \\ 0 & 2.19969625 \end{pmatrix}$$

$$0.6326829902$$

$$\begin{pmatrix} 0.3666160417 & 0 \\ 0 & -2.19969625 \end{pmatrix}$$

```
float(nsol[2]*360/(2*PI))
```

36.25006511

```
dmin := -2: dmax := 2:
vf := plot::VectorField2d(f, x=dmin..dmax, y=dmin..dmax, Color=RGB::Black):
pp := plot::Streamlines2d(f, x=dmin..dmax, y=dmin..dmax, MinimumDistance=0.1, Color=RGB::Bl
eq := plot::Point2d(0,0, Color=RGB::Red, PointSize=2):
sep := plot::Implicit2d(y^2-(2/3)*x^3, x=dmin..dmax, y=dmin..dmax, Color=RGB::Red):
```

```
plot(vf,pp,eq,sep,
     ViewingBox=[dmin..dmax,dmin..dmax], Scaling=Constrained, TicksNumber=None)
```



Richtungsaufblasungen

positive x-Richtung

```
a := 2: b:=3:
Psi_xp := [ xx^a, yy*xx^b ];
SPsi_xp := [ x=xx^a, y=yy*xx^b ]:
J_xp := jacobian(Psi_xp, [xx,yy]):
JI_xp := simplify(J_xp^(-1)):
```

$[xx^2, xx^3\, yy]$

```
g := subs(simplify(JI_xp*subs(f,SPsi_xp)), [xx=x,yy=y]); g := g/x;
sol := solve(g,[x,y]); sol1 := sol[1]: sol2 :=sol[2]:
J := linalg::jacobian(g, [x,y]):
sol1; subs(J, sol1);
sol2; subs(J, sol2)
```

$$\begin{pmatrix} \frac{x^2\, y}{2} \\ -\frac{x\,(3\,y^2-2)}{2} \end{pmatrix}$$

$$\begin{pmatrix} \frac{x\, y}{2} \\ 1-\frac{3\,y^2}{2} \end{pmatrix}$$

$$\left\{\left[x=0,\ y=-\frac{\sqrt{6}}{3}\right],\ \left[x=0,\ y=\frac{\sqrt{6}}{3}\right]\right\}$$

$$\left[x=0,\ y=-\frac{\sqrt{6}}{3}\right]$$

$$\begin{pmatrix} -\frac{\sqrt{6}}{6} & 0 \\ 0 & \sqrt{6} \end{pmatrix}$$

$$0,\ -\frac{\sqrt{6}}{3}$$

$$\left[x=0,\ y=\frac{\sqrt{6}}{3}\right]$$

$$\begin{pmatrix} \frac{\sqrt{6}}{6} & 0 \\ 0 & -\sqrt{6} \end{pmatrix}$$

```
c := map(sol2,op,2)[2];
sepeq := y^a - c^a*x^b
```

$$\frac{\sqrt{6}}{3}$$

$$y^2 - \frac{2\,x^3}{3}$$

```
dmin := -2: dmax := 2:
vf := plot::VectorField2d(g, x=dmin..dmax, y=dmin..dmax, Color=RGB::Black):
pp := plot::Streamlines2d(g, x=dmin..dmax, y=dmin..dmax, MinimumDistance=0.1, Color=RGB::Bl
eq1 := plot::Point2d(op(map(sol1,op,2)), Color=RGB::Red, PointSize=2):
eq2 := plot::Point2d(op(map(sol2,op,2)), Color=RGB::Red, PointSize=2):
```

```
plot(vf,pp,eq1,eq2,
     ViewingBox=[dmin..dmax,dmin..dmax], Scaling=Constrained, TicksNumber=None)
```



negative x-Richtung

```
a := 2: b:=3:
Psi_xn := [ -xx^a, yy*xx^b ];
SPsi_xn := [ x=-xx^a, y=yy*xx^b ]:
J_xn := jacobian(Psi_xn, [xx,yy]):
JI_xn := simplify(J_xn^(-1)):
```

$$[-xx^2, xx^3\ yy]$$

```
g := subs(simplify(JI_xn*subs(f,SPsi_xn)), [xx=x,yy=y]); g := g/x;
sol := solve(g,[x,y]);
```

$$\begin{pmatrix} -\frac{x^2\,y}{2} \\ \frac{x\,(3\,y^2+2)}{2} \end{pmatrix}$$

$$\begin{pmatrix} -\frac{x\,y}{2} \\ \frac{3\,y^2}{2} + 1 \end{pmatrix}$$

$$\left\{\left[x = 0,\ y = -\frac{\sqrt{6}\,\mathrm{i}}{3}\right], \left[x = 0,\ y = \frac{\sqrt{6}\,\mathrm{i}}{3}\right]\right\}$$

positive y-Richtung (negative y-Richtung nicht nötig, da b ungerade)

```
a := 2: b:=3:
Psi_yp := [ xx*yy^a, yy^b ];
SPsi_yp := [ x=xx*yy^a, y=yy^b ]:
J_yp := jacobian(Psi_yp, [xx,yy]):
JI_yp := simplify(J_yp^(-1)):
```

$$[xx\ yy^2, yy^3]$$

```
g := subs(simplify(JI_yp*subs(f,SPsi_yp)), [xx=x,yy=y]); g := g/y;
sol := solve(g,[x,y]); sol := sol[3]:
J := linalg::jacobian(g, [x,y]):
J := subs(J, sol)
```

$$\begin{pmatrix} -\frac{y\left(2x^3-3\right)}{3} \\ \frac{x^2 y^2}{3} \end{pmatrix}$$

$$\begin{pmatrix} 1-\frac{2x^3}{3} \\ \frac{x^2 y}{3} \end{pmatrix}$$

$$\left\{\left[x = \frac{2^{2/3}\,3^{1/3}\left(-\frac{1}{2}+\sigma_1\right)}{2},\ y=0\right],\ \left[x = -\frac{2^{2/3}\,3^{1/3}\left(\frac{1}{2}+\sigma_1\right)}{2},\ y=0\right],\ \left[x = \frac{2^{2/3}\,3^{1/3}}{2},\ y=0\right]\right\}$$

where

$$\sigma_1 = \frac{\sqrt{3}\,i}{2}$$

$$\begin{pmatrix} -2^{1/3}\,3^{2/3} & 0 \\ 0 & \frac{2^{1/3}\,3^{2/3}}{6} \end{pmatrix}$$

```
dmin := -2: dmax := 2:
vf := plot::VectorField2d(g, x=dmin..dmax, y=dmin..dmax, Color=RGB::Black):
pp := plot::Streamlines2d(g, x=dmin..dmax, y=dmin..dmax, MinimumDistance=0.1, Color=RGB::Bl
eq := plot::Point2d(op(map(sol,op,2)), Color=RGB::Red, PointSize=2):
```

```
plot(vf,pp,eq,
     ViewingBox=[dmin..dmax,dmin..dmax], Scaling=Constrained, TicksNumber=None)
```




```
c := map(sol,op,2)[1];
sepeq := c^b*y^a - x^b
```

$$\frac{2^{2/3}\,3^{1/3}}{2}$$

$$\frac{3y^2}{2} - x^3$$